

How to Format & Restore Full SD Card Capacity

Complete Diskpart guide for removing locked EFI/OS partitions and restoring raw storage

⚠ Important Safety Warning: Diskpart is a powerful low-level disk management utility. Cleaning the wrong disk will erase all partitions on that drive. Always double check your disk size and number before running the clean command.

Understanding the Problem

When an SD card is used for bootable operating systems, embedded firmware, or multi-partition setups, it gets divided into hidden or EFI partitions. Standard Windows File Explorer formatting only interacts with the active visible partition (often showing as 16 MB or 300 MB instead of the true 32 GB / 64 GB capacity). **Diskpart** bypasses Windows Explorer restrictions by completely wiping the partition table.

Step-by-Step Diskpart Instructions

1 Open Command Prompt as Administrator

Press **Win + R**, type `cmd`, and press **Ctrl + Shift + Enter** (or search for Command Prompt and select *Run as administrator*).

2 Launch the Diskpart Utility

In the command prompt window, start the disk partitioning tool:

```
diskpart
```

3 Identify Your SD Card

List all connected storage devices to identify the disk number of your SD card (verify by disk size, e.g., 58 GB):

```
list disk
```

4 Select the SD Card

Replace X with your actual SD card disk number (e.g., Disk 1):

```
select disk 1
```

5 Wipe All Existing Partitions

Remove all partition tables, EFI systems, and file structures from the card:

```
clean
```

6 Create a New Primary Partition

Create a single unallocated space partition utilizing 100% of the SD card's storage:

```
create partition primary
```

7 Select the New Partition

Explicitly focus on the newly created primary partition to prepare it for formatting:

```
select partition 1
```

8 Format the Drive

Format the partition with the chosen file system (**exFAT** for ≥ 64 GB, or **FAT32** for ≤ 32 GB):

```
format fs=exfat quick
```

9 Assign Drive Letter & Exit

Mount the partition so it shows up in Windows Explorer, then exit Diskpart:

```
assign  
exit
```

File System Selection Guide

FILE SYSTEM	RECOMMENDED SIZE	MAX FILE SIZE	BEST USED FOR
exFAT	64 GB and above	16 EB (Virtually unlimited)	Modern SD cards, 4K video, smartphones, Mac & Windows cross-compatibility.
FAT32	32 GB and below	4 GB	Microcontrollers, dashcams, 3D printers, legacy hardware.
NTFS	Any size (Internal)	16 TB	Windows internal storage and system drives (read-only on Mac by default).

 Summary Cheat Sheet:

diskpart → list disk → select disk [N] → clean → create partition primary → select partition 1 → format fs=exfat quick → assign → exit